Práctica 3

Representación de Superficies con ordenador

3.1. Enunciados de las Prácticas

Práctica A Del listado de superficies que viene a continuación sabemos que el punto $(0,0)$ es un punto crítico. En esta práctica utilizamos el software matemático para obtener sus representaciones gráficas y comprobar que el carácter que el desarrollo de Taylor asigna al punto crítico $(0,0)$ está conforme con el que se visualiza en el ordenador:

$$\begin{array}{ll}
z_1 = -(x^2+y^2)e^{-2(x^2+y^2)}; & z_2 = 2x^4+2y^4-x^2+2xy-y^2; \\
z_3 = 8xy(1-x^2-2y^2)^{1/2}; & z_4 = x^2+y^3; \\
z_5 = x^2-xy^2+y^8; & z_6 = x^2-xy^2-y^4; \\
z_7 = x^2-xy^2+y^4; & z_8 = (x^2+y^2)^{2/3}; \\
z_9 = (1+e^y)\cos(x)-ye^y; & z_{10} = x^2+y^4; \\
z_{11} = \cos(10x^2)+\cos(10y^2); & z_{12} = \operatorname{sen}(6xy);
\end{array}$$

Práctica B Consideramos el recinto cerrado y acotado del plano $K=[-1,1]\times[-1,1]$ y en K la función real $f(x,y)$ definida por la igualdad:

$$f(x,y) = \frac{\operatorname{sen}(1+2x^2+y^2)}{5+x+y}$$

Se pide:

1. Calcular todos los puntos críticos de f en K y determinar su carácter.
2. Determinar el máximo y el mínimo absoluto de f en K y donde se producen.

Práctica C Repetir el estudio de la práctica anterior para la función

$$g(x,y) = \frac{\operatorname{sen}(1+x^2+y^2)}{3+x+y}$$



Figura 3.1: Representación de las funciones $h_1(x), h_2(x)$ cuando $-3 \leq x \leq 3$.

Práctica puntuable D Comprobar que la condición necesaria para $C \in R^2$ sea un punto crítico de la función $f\colon \mathbb{R}^2 \to \mathbb{R}$, f la función de la práctica B:

$$f(x,y) = \frac{\operatorname{sen}(1+2x^2+y^2)}{5+x+y}$$

es que $C = (\overline{x}, 2\overline{x})$ para $\overline{x}$ cualquiera de los infinitos puntos en los que las funciones

$$h_1(x) = \cos(1+x^2), \qquad h_2(x) = \frac{\operatorname{sen}(1+6x^2)}{20x+12x^2}$$

toman el mismo valor. La representación de $h_1(x)$ y $h_2(x)$ cuando $-3 \leq x \leq 3$ se encuentra en la figura 3.1.

Por último, de la igualdad $\lim_{x\to\infty} h_2(x) = 0$ deducir la igualdad asintótica

$$\overline{x} \approx \pm\sqrt{\frac{(2k+1)\pi - 2}{12}}, \qquad \text{para } k \in \mathbb{N} \text{ y } k >> 0$$

3.2. Elementos necesarios

Cargamos los elementos necesarios para realizar la práctica

```
> with(plots):
```

```
> with(linalg):
```

```
> macro(kkk=(style=patchcontour));
```

$$kkk$$

3.3. Primera superficie de la práctica A

```
> z:=-(x^2+y^2)*exp(-2*(x^2+y^2));
> Ho:=subs({x=0,y=0},hessian(z,[x,y]));
> plot3d(z,x=-1..1,y=-1..1,orientation=[85,75],kkk);
> H1:=subs({x=1/sqrt(2),y=0},hessian(z,[x,y]));
```

$$z := -(x^2+y^2)\,e^{(-2\,x^2-2\,y^2)}$$

$$Ho := \begin{bmatrix} -2\,e^0 & 0 \\ 0 & -2\,e^0 \end{bmatrix}$$



$$H1 := \begin{bmatrix} 4\,e^{(-1)} & 0 \\ 0 & 0 \end{bmatrix}$$

z_1 tiene en (0,0) un máximo y en $(1/\sqrt{2}, 0)$ un quasi-mínimo que es un mínimo

3.4. Segunda superficie de la práctica A

```
> z:=2*x^4+2*y^4-x^2+2*x*y-y^2;
> Ho:=subs({x=0,y=0},hessian(z,[x,y]));
> plot3d(z,x=-1..1,y=-1..1,orientation=[-162,106],kkk);
```

$$z := 2\,x^4 + 2\,y^4 - x^2 + 2\,x\,y - y^2$$

$$Ho := \begin{bmatrix} -2 & 2 \\ 2 & -2 \end{bmatrix}$$



En (0,0) z_2 tiene un quasi-máximo que es un punto de ensilladura

3.5. Tercera superficie de la práctica A

```
> z:=8*x*y*sqrt(1-x^2-2*y^2);
> zz:=8*x*y*sqrt(max(0,1-x^2-2*y^2)):
> Ho:=subs({x=0,y=0},hessian(z,[x,y]));
> plot3d(zz,x=-1..1,y=-1..1,orientation=[-83,47],kkk);
```

$$z := 8\,x\,y\,\sqrt{1 - x^2 - 2\,y^2}$$

$$Ho := \begin{bmatrix} 0 & 8 \\ 8 & 0 \end{bmatrix}$$



En (0,0) z_3 tiene una ensilladura. Representamos zz por problemas en el dominio.

3.6. Cuarta superficie de la práctica A

```
> z:=x^2+y^3;
> Ho:=subs({x=0,y=0},hessian(z,[x,y]));
> plot3d(z,x=-1..1,y=-1..1,orientation=[-59,68],kkk);
```

$$z := x^2 + y^3$$

$$Ho := \begin{bmatrix} 2 & 0 \\ 0 & 0 \end{bmatrix}$$



En (0,0) z_4 tiene un quasi-mínimo que es un punto de ensilladura.

3.7. Quinta superficie de la práctica A

```
> z:=x^2-x*y^2+y^8;
> Ho:=subs({x=0,y=0},hessian(z,[x,y]));
> plot3d(z,x=-1..1,y=-1..1,orientation=[33,38],kkk);
```

$$z := x^2 - x\,y^2 + y^8$$

$$Ho := \begin{bmatrix} 2 & 0 \\ 0 & 0 \end{bmatrix}$$



En (0,0) z_5 tiene un quasi-mínimo que es un punto de ensilladura.

3.8. Sexta superficie de la práctica A

```
> z:=x^2-x*y^2-y^4;
> Ho:=subs({x=0,y=0},hessian(z,[x,y]));
> plot3d(z,x=-1..1,y=-1..1,orientation=[59,31],kkk);
```

$$z := x^2 - x\,y^2 - y^4$$

$$Ho := \begin{bmatrix} 2 & 0 \\ 0 & 0 \end{bmatrix}$$



En (0,0) z_6 tiene un quasi-mínimo que es un punto de ensilladura.

3.9. Séptima superficie de la práctica A

```
> z:=x^2-x*y^2+y^4;
> Ho:=subs({x=0,y=0},hessian(z,[x,y]));
> plot3d(z,x=-1..1,y=-1..1,orientation=[37,50],kkk);
```

$$z := x^2 - x\,y^2 + y^4$$

$$Ho := \begin{bmatrix} 2 & 0 \\ 0 & 0 \end{bmatrix}$$



En (0,0) z_7 tiene un quasi-mínimo que es un mínimo.

3.10. Octava superficie de la práctica A

```
> z:=(x^2+y^2)^(3/2);
> Taylor_z:=mtaylor(z,[x,y],3);
> plot3d(z,x=-1..1,y=-1..1,orientation=[26,99],kkk);
```

$$z := (x^2 + y^2)^{(3/2)}$$

$$Taylor\_z := 0$$



En (0,0) z_8 tiene un extremo que es un mínimo.

3.11. Novena superficie de la práctica A

```
> z:=(1+exp(y))*cos(x)-y*exp(y);
> Ho:=subs({x=0,y=0},hessian(z,[x,y])): Ho:=evalf(%);
> plot3d(z,x=-1..1,y=-1..1,orientation=[-29,82],kkk);
```

$$z := (1 + e^y)\cos(x) - y\,e^y$$

$$Ho := \begin{bmatrix} -2. & -0. \\ -0. & -1. \end{bmatrix}$$



En (0,0) z_9 tiene un máximo.

3.12. Décima superficie de la práctica A

```
> z:=x^2+y^4;
> Ho:=subs({x=0,y=0},hessian(z,[x,y]));
> plot3d(z,x=-1..1,y=-1..1,orientation=[30,102],kkk);
```

$$z := x^2 + y^4$$

$$Ho := \begin{bmatrix} 2 & 0 \\ 0 & 0 \end{bmatrix}$$



En (0,0) z_{10} tiene un quasi-mínimo que es un mínimo.

3.13. Undécima superficie de la práctica A

```
> z:=cos(10*x^2)+cos(10*y^2);
> Ho:=subs({x=0,y=0},hessian(z,[x,y])): Ho:=evalf(%);
> plot3d(z,x=-1..1,y=-1..1,orientation=[113,52],kkk);
```

$$z := \cos(10\,x^2) + \cos(10\,y^2)$$

$$Ho := \begin{bmatrix} -0. & 0. \\ 0. & -0. \end{bmatrix}$$



En (0,0) z_{11} tiene un extremo de matriz hessiana nula que es un máximo.

3.14. Duodécima superficie de la práctica A

```
> z:=sin(6*x*y);
> Ho:=subs({x=0,y=0},hessian(z,[x,y])): Ho:=evalf(%);
> plot3d(z,x=-1..1,y=-1..1,orientation=[98,39],kkk);
```

$$z := \sin(6\,x\,y)$$

$$Ho := \begin{bmatrix} 0. & 6. \\ 6. & 0. \end{bmatrix}$$



En (0,0) z_{12} tiene un punto de ensilladura.

3.15. Observaciones a lo obtenido en la práctica A

Como sabemos por la teoría un quasi-mínimo es un mínimo respecto a todas las direcciones excepto, posiblemente, a una. Con las superficies z_4, z_5, z_6, z_7 podemos observar que todas tienen el mismo desarrollo de Taylor de orden 3 en $P_0 = (0,0)$ de lo que se deduce que el carácter de P_0 es, en todas ellas, un quasi-mínimo que es un mínimo respecto a todas las direcciones excepto, posiblemente, la dada por la recta $r: y = x$. Ahora, respecto a la recta r tenemos los siguientes comportamientos:

1. z_4 en P_0 y en la dirección r posee una inflexión. Esto implica que el carácter definitivo de P_0 es el de un punto de ensilladura.

2. z_5 en P_0 y en la dirección r posee un mínimo. Esto implica que z_5 respecto a todas las direcciones es un mínimo y, sin embargo, su carácter definitivo no es el de un mínimo relativo sino el de un punto de ensilladura, tal y como comprobamos a partir de su figura y sus curvas de nivel.

3. z_6 en P_0 y en la dirección r posee un máximo. Esto implica que el carácter definitivo de P_0 es el de un punto de ensilladura.

4. z_7 en P_0 y en la dirección r posee un mínimo. De la figura concluimos que el carácter definitivo de P_0 es el de mínimo relativo.

3.16. Solución de la práctica B

```
> f:=sin(1+2*x^2+y^2)/(5+x+y);  # f expresión
> F:=unapply(f,(x,y));      # F es la función f
```

$$f := \frac{\sin(1+2\,x^2+y^2)}{5+x+y}$$

$$F := (x,\,y) \to \frac{\sin(1+2\,x^2+y^2)}{5+x+y}$$

```
> diff(f,x): fx:=simplify(%);
> diff(f,y): fy:=simplify(%);
```

$$fx := \frac{20\cos(\%1)\,x + 4\cos(\%1)\,x^2 + 4\cos(\%1)\,x\,y - \sin(\%1)}{(5+x+y)^2}$$

$$\%1 := 1+2\,x^2+y^2$$

$$fy := \frac{10\cos(\%1)\,y + 2\cos(\%1)\,x\,y + 2\cos(\%1)\,y^2 - \sin(\%1)}{(5+x+y)^2}$$

$$\%1 := 1+2\,x^2+y^2$$

```
> D1:=implicitplot(fx=0,x=-1..1,y=-1..1,numpoints=5000):
> D2:=implicitplot(fy=0,x=-1..1,y=-1..1,numpoints=5000,color=blue):
> display(D1,D2); r1:=fsolve({fx,fy},{x,y},x=-1..0, y=-1..0);
> r2:=fsolve({fx,fy},{x=0.1,y=0.2});
> r3:=fsolve({fx,fy},{x=0.3,y=0.4});
```



Figura 3.2: Cálculo de los puntos críticos de f,
el lugar $\frac{\partial f}{\partial x} = 0$ está resaltado con una línea más gruesa.

$$r1 := \{x = -0.3532064078,\ y = -0.7064128156\}$$
$$r2 := \{x = 0.08113279991,\ y = 0.1622655998\}$$
$$r3 := \{x = 0.2601448679,\ y = 0.5202897357\}$$

```
> plot3d(f,x=-1..1,y=-1..1);
```



Figura 3.3: Representación gráfica de f.

```
> H1:=subs(r1,hessian(f,[x,y])):
> H1:=evalf(%):
> print(`H1=`,H1,det(H1),` Máximo`);
> H2:=subs(r2,hessian(f,[x,y])): H2:=evalf(%):
> print(`H2=`,H2,det(H2),` Mínimo`);
> H3:=subs(r3,hessian(f,[x,y])):H3:=evalf(%):
> print(`H3=`,H3,det(H3),` Silla`);
```

$$H1 =, \begin{bmatrix} -0.6780625102 & -0.4985894744 \\ -0.4985894744 & -0.5883259924 \end{bmatrix}, 0.1503303352, \mathit{Máximo}$$

$$H2 =, \begin{bmatrix} 0.3691919090 & -0.01731737917 \\ -0.01731737917 & 0.1759372649 \end{bmatrix}, 0.06465472307, \mathit{Mínimo}$$

$$H3 =, \begin{bmatrix} -0.0712999613 & -0.1847862464 \\ -0.1847862464 & -0.1280431037 \end{bmatrix}, -0.02501648852, \mathit{Silla}$$

```
> F(-1,-1.); F(-1,1.); F(1,-1.); F(1,1.);
```

$$-0.2522674984$$
$$-0.1513604991$$
$$-0.1513604991$$
$$-0.1081146422$$

3.16.1. Conclusiones de los cálculos de la práctica B

De la figura 3.2 se deduce que la función f en $K = [-1,1] \times [-1,1]$ tiene exactamente tres puntos críticos

$$p_1 = (x = -0.3532064078,\ y = -0.7064128156)$$
$$p_2 = (x = 0.08113279991,\ y = 0.1622655998)$$
$$p_3 = (x = 0.2601448679,\ y = 0.5202897357)$$

ahora, de la figura 3.3 y de los Hessianos $H1$, $H2$ y $H3$, se deduce:

1. p_1 es el único máximo relativo de f en K y también es su máximo absoluto.
2. p_2 es el único mínimo relativo de f en K y no es su mínimo absoluto.
3. p_3 es el único punto de ensilladura de f en K.

Por último, de la figura 3.3, deducimos que el mínimo absoluto de f en K se alcanza en uno de los puntos $(-1,-1),(-1,1),(1,-1),(1,1)$, en el último cálculo realizado comprobamos que $(-1,-1)$ es el mínimo absoluto de f en K.

3.17. Solución de la práctica C

```
> g:=sin(1+x^2+y^2)/(3+x+y);
> G:=unapply(f,(x,y));   # F es la función f
```

$$g := \frac{\sin(1+x^2+y^2)}{3+x+y}$$
$$G := (x,\, y) \to f$$

```
> diff(g,x): gx:=simplify(%);
> diff(g,y): gy:=simplify(%);
```

$$gx := (6\cos(1+x^2+y^2)\,x + 2\cos(1+x^2+y^2)\,x^2 + 2\cos(1+x^2+y^2)\,x\,y$$
$$-\sin(1+x^2+y^2))\,/(3+x+y)^2$$

$$gy := (6\cos(1+x^2+y^2)\,y + 2\cos(1+x^2+y^2)\,x\,y + 2\cos(1+x^2+y^2)\,y^2$$
$$-\sin(1+x^2+y^2))\,/(3+x+y)^2$$

```
> D1:=implicitplot(gx=0,x=-1..1,y=-1..1,numpoints=5000):
> D2:=implicitplot(gy=0,x=-1..1,y=-1..1,numpoints=5000,color=blue):
> display(D1,D2); r1:=fsolve({gx,gy},{x,y},x=-1..0);
```



$$r1 := \{x = -0.7037181428,\ y = -0.7037181428\}$$

```
> plot3d(g,x=-1..1,y=-1..1,orientation=[10,75],kkk);
```



```
> H1:=subs(r1,hessian(f,[x,y])): H1:=evalf(%); det(H1);
```

$$H1 := \begin{bmatrix} -1.647578064 & -1.135907921 \\ -1.135907921 & -1.647578064 \end{bmatrix}$$

$$1.424226672$$

```
> F(-1,-1.); F(-1,1.); F(1,-1.); F(1,1.);
```

$$0.1411200081$$
$$0.04704000270$$
$$0.04704000270$$
$$0.02822400162$$

El único punto crítico de la función g en K es r_1. r_1 es un máximo relativo que es máximo absoluto y (1,1) es el mínimo absoluto de g en K.

Práctica 4

Movimiento Armónico

4.1. Enunciado de la Práctica

Ver [AKLo], vol. I, p. 377.

Una partícula de masa 0.5 $[\![\text{kg}]\!]$ se mueve sobre el eje horizontal Ox en un medio resistente o viscoso (por ejemplo, un líquido o un gas) bajo la influencia de la fuerza elástica de dos resortes (ver figura 4.1) que actúan según la ley de Hooke; esta ley establece que la fuerza elástica está dirigida hacia la posición de equilibrio y es proporcional a la desviación a partir de dicha posición de equilibrio. Supongamos que la posición de equilibrio equivale a $x = 0$ y que la fuerza elástica es $-kx\,[\![\text{N}]\!]$ $(k > 0)$.

Figura 4.1: Partícula bajo la acción de la fuerza elástica de dos resortes.

Además, la resistencia del medio es proporcional a la velocidad de la partícula e igual a $-\lambda \dfrac{dx}{dt}\,[\![\text{N}]\!]$ $(\lambda > 0)$.

Sabiendo que la posición inicial de la partícula es la de reposo en $x = 4$, se pide determinar:

a) La ecuación diferencial que describe el movimiento cuando la partícula está sometida, además, a la acción de una fuerza externa de magnitud $E(t)\,[\![\text{N}]\!]$.

b) La posición $x(t)$, t en segundos, de la partícula en el caso $\lambda = 0,\ k = 2, E(t) = 0$.

c) La posición $x(t)$ de la partícula en el caso $\lambda = 0.5,\ k = 2.125, E(t) = 0$.

d) La posición $x(t)$ de la partícula en el caso $\lambda = 0, k = 9, E(t) = 40 \cdot \cos(5t)$.

e) La posición $x(t)$ de la partícula en el caso $\lambda = 0, k = 9, E(t) = 2\cos 2t$.

f) Para $t \in [0, 10]$, calcular la máxima amplitud del movimiento en el caso d).

g) ¿En qué instantes alcanza la partícula la posición de equilibrio en los casos b) y c)?

4.2. Ecuación del movimiento

Teniendo en cuenta que la fuerza a la que se ve sometida nuestra partícula coincide con el producto de su masa por su aceleración, i.e., $F = m \cdot \dfrac{d^2x}{dt^2}$, resulta que:

$$m \cdot \frac{d^2x}{dt^2} = -kx - \lambda \frac{dx}{dt} \tag{4.1}$$

de lo que resulta que la ecuación diferencial que describe el desplazamiento, $x(t)$, de nuestra partícula en función del tiempo t tiene la forma

$$x'' + \frac{\lambda}{m}x' + \frac{k}{m}x = 0, \qquad \lambda \geq 0, k \geq 0 \tag{4.2}$$

Esta ecuación se transforma en la más general

$$x'' + \frac{\lambda}{m}x' + \frac{k}{m}x = \frac{E(t)}{m}, \qquad \lambda \geq 0, k \geq 0 \tag{4.3}$$

en aquellos casos en que además nuestra partícula se encuentra sometida a una fuerza exterior de $E(t)$ $[\![\mathrm{N}]\!]$.

4.3. Elementos necesarios

```
> restart;
> with(plots):
```

```
Warning, the name changecoords has been redefined
```

4.4. Apartado b: Movimiento armónico simple

```
> ecB:=(D@@2)(x)(t)+4*x(t)=0;
> ini:=x(0)=4,D(x)(0)=0; dsolve({ecB,ini},x(t)); xB:=rhs(%);
> plot(xB,t=0..20);
```

$$ecB := (\mathrm{D}^{(2)})(x)(t) + 4\,\mathrm{x}(t) = 0$$

$$ini := \mathrm{x}(0) = 4,\ \mathrm{D}(x)(0) = 0$$

$$\mathrm{x}(t) = 4\cos(2\,t)$$

$$xB := 4\cos(2\,t)$$



Figura 4.2: Movimiento armónico simple

4.5. Apartado c: Movimiento armónico amortiguado

```
> ecC:=(D@@2)(x)(t)+D(x)(t)+(425/100)*x(t)=0;
> dsolve({ecC,ini},x(t)): xC:=rhs(%);
> plot(xC,t=0..20);
```

$$ecC := (\mathrm{D}^{(2)})(x)(t) + \mathrm{D}(x)(t) + \frac{17}{4}\,\mathrm{x}(t) = 0$$

$$xC := e^{(-\frac{t}{2})}\sin(2\,t) + 4\,e^{(-\frac{t}{2})}\cos(2\,t)$$



Figuras 4.3: Movimiento armónico amortiguado

4.6. Apartados d,f: Movimiento armónico forzado

```
> ecD:=(D@@2)(x)(t)+18*x(t)=80*cos(5*t);
> dsolve({ecD,ini},x(t)): xD:=rhs(%);
> plot(xD,t=0..20);
```

$$ecD := (\mathrm{D}^{(2)})(x)(t) + 18\,\mathrm{x}(t) = 80\cos(5\,t)$$

$$xD := \frac{108}{7}\cos(3\sqrt{2}\,t) - \frac{80}{7}\cos(5\,t)$$

Figura 4.4: Movimiento armónico con interferencia

Vamos ahora a comprobar que la máxima amplitud del caso d) cuando $t \in [0, 10]$ es $A_m = 26.71686747$ ⟦m⟧ para ello utilizamos el listado:

```
> plot(abs(xD),t=0..10);
```




```
> dxD:=diff(xD,t); t0:=fsolve(dxD,{t=4.3});
> Am:=subs(t0,xD);
> Am:=evalf(Am);
```

$$dxD := -\frac{324}{7}\sin(3\sqrt{2}\,t)\,\sqrt{2} + \frac{400}{7}\sin(5\,t)$$

$$t0 := \{t = 4.420244699\}$$

$$Am := \frac{108}{7}\cos(13.26073410\,\sqrt{2}) - \frac{80}{7}\cos(22.10122350)$$

$$Am := 26.71686747$$

De la representación gráfica de $|x_D(t)|$, $t \in [0, 10]$, vemos que el máximo se consigue cerca de $t = 4.3$, la exactitud la conseguimos con la instrucción `t0:=fsolve(dxD,{t=4.3});` y el cálculo de la amplitud máxima con las dos últimas instrucciones.

4.7. Apartado e: Movimiento armónico en resonancia

```
> ecE:=(D@@2)(x)(t)+4*x(t)=4*cos(2*t);
> dsolve({ecE,ini},x(t)): xE:=rhs(%);
> plot(xE,t=0..20);
```

$$ecE := (\mathrm{D}^{(2)})(x)(t) + 4\,\mathrm{x}(t) = 4\cos(2\,t)$$

$$xE := 4\cos(2\,t) + \sin(2\,t)\,t$$



Figura 4.5: Movimiento armónico en resonancia

Según podemos observar por el gráfico, y con mayor certeza a partir de la expresión de $x_E(t)$, la amplitud de $x_E(t)$ aumenta según lo hace t y, por ello, su amplitud máxima es infinita.

4.8. Solución al apartado g

Allí donde la x se hace cero. Tal como hemos visto en el anterior apartado se verifica que las soluciones $x_B(t)$, $x_C(t)$ de los apartados b) y c) tienen, respectivamente, la expresión

$$x_B(t) = 4\cos 2t; \qquad x_C(t) = e^{-t/2}(4\cos 2t + \operatorname{sen} 2t) \tag{4.4}$$

teniendo en cuenta la igualdad $\operatorname{sen}(\alpha + 2t) = \operatorname{sen}\alpha\cos(2t) + \cos\alpha\operatorname{sen}(2t)$ podemos poner

$$\begin{aligned} 4\cos 2t + \operatorname{sen} 2t &= \sqrt{17}(\tfrac{4}{\sqrt{17}}\cos 2t + \tfrac{1}{\sqrt{17}}\operatorname{sen} 2t) \\ &= \sqrt{17}\operatorname{sen}(\alpha + 2t) \end{aligned} \tag{4.5}$$

a condición de que $\alpha = \operatorname{arctg}(4) = 1.3258$, con lo cual

$$x_B(t) = 4\cos 2t; x_C(t) = \sqrt{17}e^{-t/2}(\operatorname{sen}(\alpha + 2t)) \tag{4.6}$$

y ahora ya es fácil observar que las posiciones de equilibrio se consiguen, respectivamente, en los puntos

$$\begin{aligned} P_B &= \{\tfrac{\pi}{4} + k\tfrac{\pi}{2};\quad k \geq 0\} = \{0.7854 + k\tfrac{\pi}{2};\quad k \geq 0\} \\ P_C &= \{\tfrac{k\pi - 1.3258}{2};\quad k \geq 1\} = \{0.9079 + k\tfrac{\pi}{2};\quad k \geq 0\} \end{aligned} \tag{4.7}$$

4.9. Animaciones

Damos como ejemplo las instrucciones necesarias para producir la animación del movimiento de la partícula sometida a las condiciones del caso b):

```
> XA:=unapply(xA,t);
```

$$XA := t \to 4\cos(2\,t)$$

```
> for i from 0 to 200 do
> p[i]:=plot([[XA(0.1*i),0]],x=-4.5..4.5,style=point,symbol=circle)
> end do:
> display([seq(p[i],i=0..200)],insequence=true);
```

4.10. Espacio de Fases y de Flujo

La ecuación diferencial ecB: $\mathrm{D}^2x + 4x = 0$ se puede poner en su forma equivalente:

$$ECb\colon\ \begin{cases} \mathrm{D}x &= v \\ \mathrm{D}v &= -4x \end{cases} \tag{4.8}$$

Ahora la representación del campo vectorial $X = v\frac{\partial}{\partial x} - 4x\frac{\partial}{\partial v} \equiv (v, -4x)$ y la pintura de la solución de ecB y de otras dos equivalentes en el espacio de fases (X,V) nos queda:

```
> with(DEtools):

> ECb:=[D(x)(t)=v(t),
> D(v)(t)=-4*x(t)];
```

```
> DEplot(ECb,[x(t),v(t)],
> t=0..20,x=-5..5,v=-10..10, [[x(0)=4,v(0)=0],[x(0)=3,v(0)=0],
> [x(0)=2,v(0)=0]]);
```

$$ECb := [\mathrm{D}(x)(t) = \mathrm{v}(t),\, \mathrm{D}(v)(t) = -4\,\mathrm{x}(t)]$$



Haciendo lo análogo en el caso ecC nos queda:

```
> ECc:=[D(x)(t)=v(t),
> D(v)(t)=-D(x)(t)-(425/100)*x(t)]; DEplot(ECc,[x(t),v(t)],t=0..20,
> x=-5..5,v=-6..6,[[x(0)=4,v(0)=0],
> [x(0)=3,v(0)=0],[x(0)=2,v(0)=0]]);
```

$$ECc := [\mathrm{D}(x)(t) = \mathrm{v}(t),\, \mathrm{D}(v)(t) = -\mathrm{D}(x)(t) - \frac{17}{4}\,\mathrm{x}(t)]$$